
Overview: RELIABILITY Procedure
The RELIABILITY procedure provides tools for reliability and survival data analysis and for recurrent events
data analysis. You can use this procedure to

� construct probability plots and fitted life distributions with left-censored, right-censored, and interval-
censored lifetime data

� fit regression models, including accelerated life test models, to combinations of left-censored, right-
censored, and interval-censored data

� analyze recurrence data from repairable systems

These tools benefit reliability engineers and industrial statisticians working with product life data and system
repair data. They also aid workers in other fields, such as medical research, pharmaceuticals, social sciences,
and business, where survival and recurrence data are analyzed.

Most practical problems in reliability data analysis involve right-censored, left-censored, or interval-censored
data. The RELIABILITY procedure provides probability plots of uncensored, right-censored, interval-
censored, and arbitrarily censored data.

Features of the RELIABILITY procedure include

� probability plotting and parameter estimation for the common life distributions: Weibull, three-
parameter Weibull, exponential, extreme value, normal, lognormal, logistic, and log-logistic. The data
can be complete, right censored, or interval censored.

� maximum likelihood estimates of distribution parameters, percentiles, and reliability functions

� both asymptotic normal and likelihood ratio con�dence intervals for distribution parameters and
percentiles. Asymptotic normal con�dence intervals for the reliability function are also available.

� estimation of distribution parameters by least squares �tting to the probability plot

� Weibayes analysis, where there are no failures and where the data analyst speci�es a value for the
Weibull shape parameter

� estimates of the resulting distribution when speci�ed failure modes are eliminated

� plots of the data and the �tted relation for life versus stress in the analysis of accelerated life test data

� �tting of regression models to life data, where the life distribution location parameter is a linear
function of covariates. The �tting yields maximum likelihood estimates of parameters of a regression
model with a Weibull, exponential, extreme value, normal, lognormal, logistic and log-logistic, or
generalized gamma distribution. The data can be complete, right censored, left censored, or interval
censored. For example, accelerated life test data can be modeled with such a regression model.

� nonparametric estimates and plots of the mean cumulative function for cost or number of recurrences
and associated con�dence intervals from data with exact or interval recurrence ages

� maximum likelihood estimation of the parameters of parametric models for recurrent events data

� horizontal plots of failure times for recurrent events data

Some of the features provided in the RELIABILITY procedure are available in other SAS procedures.

� You can construct probability plots of life data with the CAPABILITY procedure; however, the
CAPABILITY procedure is intended for process capability analysis rather than reliability analysis, and
the data must be complete (that is, uncensored).

� The LIFEREG procedure �ts regression models with life distributions such as the Weibull, lognormal,
and log-logistic to left-, right-, and interval-censored data. The RELIABILITY procedure �ts the same
distributions and regression models as the LIFEREG procedure and, in addition, provides a graphical
display of life data in probability plots.

Lawless (2003), Meeker and Escobar (1998), Nelson (1982, 1990), Abernethy (2006), and Tobias and
Trindade (1995) provide many examples taken from diverse �elds and describe the analyses provided by the
RELIABILITY procedure.

The features of the procedure that deal with the nonparametric analysis of recurrent events data from
repairable systems are based on the work of Doganaksoy and Nelson (1998), Nelson (1988, 1995, 2003), and
Nelson and Doganaksoy (1989), who provide examples of repair data analysis. Meeker and Escobar (1998),
Rigdon and Basu (2000), Cook and Lawless (2007), Abernethy (2006), Tobias and Trindade (1995), Crowder
et al. (1991), and US Army (2000) provide details of parametric models for recurrent events data.

Getting Started: RELIABILITY Procedure

This section introduces the RELIABILITY procedure with examples that illustrate some of the analyses that
it performs.

Analysis of Right-Censored Data from a Single Population

The Weibull distribution is used in a wide variety of reliability analysis applications. This example illustrates
the use of the Weibull distribution to model product life data from a single population. The following
statements create a SAS data set containing observed and right-censored lifetimes of 70 diesel engine fans
(Nelson 1982, p. 318):

data fan;
input Lifetime censor @@;
Lifetime = Lifetime/1000;
label lifetime=�Fan Life (1000s of Hours)�;
datalines;

450 0 460 1 1150 0 1150 0 1560 1
1600 0 1660 1 1850 1 1850 1 1850 1
1850 1 1850 1 2030 1 2030 1 2030 1
2070 0 2070 0 2080 0 2200 1 3000 1
3000 1 3000 1 3000 1 3100 0 3200 1
3450 0 3750 1 3750 1 4150 1 4150 1
4150 1 4150 1 4300 1 4300 1 4300 1
4300 1 4600 0 4850 1 4850 1 4850 1
4850 1 5000 1 5000 1 5000 1 6100 1
6100 0 6100 1 6100 1 6300 1 6450 1
6450 1 6700 1 7450 1 7800 1 7800 1
8100 1 8100 1 8200 1 8500 1 8500 1
8500 1 8750 1 8750 0 8750 1 9400 1
9900 1 10100 1 10100 1 10100 1 11500 1
;

Some of the fans had not failed at the time the data were collected, and the unfailed units have right-censored
lifetimes. The variableLifetime represents either a failure time or a censoring time in thousands of hours.
The variableCensor is equal to 0 if the value ofLifetime is a failure time, and it is equal to 1 if the value is a
censoring time.

If ODS Graphics is disabled, graphical output is created using traditional graphics; otherwise, ODS Graphics
is used. The following statements use the RELIABILITY procedure to produce the traditional graphical
output shown in Figure 18.1:

ODS Graphics OFF;
proc reliability data=fan;

distribution Weibull;
pplot lifetime * censor( 1 ) / covb ;

run;
ODS Graphics ON;

The DISTRIBUTION statement speci�es the Weibull distribution for probability plotting and maximum
likelihood (ML) parameter estimation. The PROBPLOT statement produces a probability plot for the variable
Lifetime and speci�es that the value of 1 for the variableCensor denotes censored observations. You can
specify any value, or group of values, for thecensor-variable(in this case,Censor) to indicate censoring
times. The option COVB requests the ML parameter estimate covariance matrix.

The graphical output, displayed in Figure 18.1, consists of a probability plot of the data, an ML �tted
distribution line, and con�dence intervals for the percentile (lifetime) values. Aninsetbox containing
summary statistics, Weibull scale and shape estimates, and other information is displayed on the plot by
default. The locations of the right-censored data values are plotted as plus signs in an area at the top of the
plot.

Figure 18.1 Weibull Probability Plot for Engine Fan Data (Traditional Graphics)

If ODS Graphics is enabled, you can create the probability plot by using ODS Graphics. The following SAS
statements use ODS Graphics to create the probability plot shown in Figure 18.1:

proc reliability data=fan;
distribution Weibull;
pplot lifetime * censor( 1 ) / covb;

run;

The plot is shown in Figure 18.2.



1222 F Chapter 18: The RELIABILITY Procedure

Figure 18.2 Weibull Probability Plot for Engine Fan Data (ODS Graphics)

The tabular output produced by the preceding SAS statements is shown in Figure 18.3 and Figure 18.4. This
consists of summary data, �t information, parameter estimates, distribution percentile estimates, standard
errors, and con�dence intervals for all estimated quantities.

Figure 18.3 Tabular Output for the Fan Data Analysis

Figure 18.3 continued

Figure 18.4 Percentile Estimates for the Fan Data

Weibull Analysis Comparing Groups of Data

This example illustrates probability plotting and distribution �tting for data grouped by the levels of a special
group-variable. The data are from an accelerated life test of an insulating �uid and are the times to electrical
breakdown of the �uid under different high voltage levels. Each voltage level de�nes a subset of data for
which a separate analysis and Weibull plot are produced. These data are the 26kV, 30kV, 34kV, and 38kV
groups of the data provided by Nelson (1990, p. 129). The following statements create a SAS data set
containing the lifetimes and voltages:

data fluid;
input Time voltage $ @@;
datalines;

5.79 26kv 1579.52 26kv
2323.7 26kv 7.74 30kv
17.05 30kv 20.46 30kv
21.02 30kv 22.66 30kv
43.4 30kv 47.3 30kv
139.07 30kv 144.12 30kv
175.88 30kv 194.90 30kv
0.19 34kv .78 34kv
0.96 34kv 1.31 34kv
2.78 34kv 3.16 34kv
4.15 34kv 4.67 34kv
4.85 34kv 6.50 34kv
7.35 34kv 8.01 34kv
8.27 34kv 12.06 34kv
31.75 34kv 32.52 34kv
33.91 34kv 36.71 34kv
72.89 34kv .09 38kv
0.39 38kv .47 38kv
0.73 38kv .74 38kv
1.13 38kv 1.40 38kv
2.38 38kv
;

The variableTime provides the time to breakdown in minutes, and the variableVoltage provides the voltage
level at which the test was conducted. These data are not censored.

The RELIABILITY procedure plots the data for the different voltage levels on the same Weibull probability
plot, �ts a separate distribution to the data at each voltage level, and superimposes distribution lines on the
plot.

The following statements produce the probability plot shown in Figure 18.5 for the variableTime at each
level of thegroup-variableVoltage:

proc reliability data=fluid;
distribution Weibull;
pplot time=voltage / overlay

noconf;
run;

The input data set FLUID is speci�ed by the DATA= option in the PROCRELIABILITY statement. The
PROBPLOT statement option OVERLAY speci�es that plots for the groups are to be overlaid rather than
displayed separately. The option NOCONF speci�es that no con�dence bands are to be plotted, since these
can interfere with one another on overlaid plots; con�dence bands are displayed by default.

Figure 18.5 Weibull Probability Plot for the Insulating Fluid Data

A summary table that contains information for all groups is displayed. In addition, information identical to
that shown in Figure 18.3 is tabulated for each level of voltage. The summary table for all groups and the
tables for the 26kV group are shown in Figure 18.6 and Figure 18.7.

Figure 18.6 Partial Listing of the Tabular Output for the Insulating Fluid Data

Figure 18.6 continued

Figure 18.7 Partial Listing of the Tabular Output for the Insulating Fluid Data

Figure 18.7 continued

Analysis of Accelerated Life Test Data

The following example illustrates the analysis of an accelerated life test for Class B electrical motor insulation.
The data are provided by Nelson (1990, p. 243). Forty insulation specimens were tested at four temperatures:
150� , 170� , 190� , and220� C. The purpose of the test is to estimate the median life of the insulation at the
design operating temperature of130� C.

The following SAS program creates the data listed in Figure 18.8. Ten specimens of the insulation were tested
at each test temperature. The variableTime provides a specimen time to failure or a censoring time, in hours.
The variableCensor is equal to 1 if the value of the variableTime is a right-censoring time and is equal to 0
if the value is a failure time. Some censor times and failure times are identical at some of the temperatures.
Rather than repeating identical observations in the input data set, the variableCount provides the number of
specimens with identical times and temperatures. The variableTemp provides the test temperature in degrees
centigrade. The variableCntrsl is a control variable specifying that percentiles are to be computed only for
the �rst value ofTemp (130� C). The value ofTemp in the �rst observation (130� C) does not correspond to
a test temperature. The missing values in the �rst observation cause the observation to be excluded from
the model �t, and the value of 1 for the variableCntrl causes percentiles corresponding to a temperature of
130� C to be computed.

data classb;
input hours temp count censor;
if _n_ = 1 then cntrl=1;
else cntrl=0;
label hours=�Hours�;
datalines;
. 130 . .

8064 150 10 1
1764 170 1 0
2772 170 1 0
3444 170 1 0
3542 170 1 0
3780 170 1 0
4860 170 1 0
5196 170 1 0
5448 170 3 1

408 190 2 0
1344 190 2 0
1440 190 1 0
1680 190 5 1

408 220 2 0
504 220 3 0
528 220 5 1

;

Figure 18.8 Listing of the Class B Insulation Data

An Arrhenius-lognormal model is �tted to the data in this example. In other words, the failure times follow
a lognormal (base 10) distribution, and the lognormal location parameter� depends on the centigrade
temperatureTemp through the Arrhenius relationship

�.x/ D � 0 C � 1x

where

x D
1000

Temp C 273:15

is 1,000 times the reciprocal absolute temperature. The lognormal (basee) distribution is also available.

The following SAS statements �t the Arrhenius-lognormal model, and they display the �tted model distribu-
tions side-by-side on the probability and the relation plots shown in Figure 18.9:

proc reliability;
distribution lognormal10;
freq count;
model hours * censor(1) = temp /

relation = arr
obstats(quantile = .1 .5 .9 control = cntrl);

rplot hours * censor(1) = temp /
pplot
fit = model
noconf
relation = arr
plotdata
plotfit 10 50 90
lupper = 1.e5
slower = 120;

run;

The PROC RELIABILITY statement invokes the procedure and speci�esCLASSB as the input data set. The
DISTRIBUTION statement speci�es that the lognormal (base 10) distribution is to be used for maximum
likelihood parameter estimation and probability plotting. The FREQ statement speci�es that the variable
Count is to be used as a frequency variable; that is, ifCount=n, then there aren specimens with the time and
temperature speci�ed in the observation.

The MODEL statement �ts a linear regression equation for the distribution location parameter as a function of
independent variables. In this case, the MODEL statement also transforms the independent variable through
the Arrhenius relationship. The dependent variable is speci�ed asTime. A value of 1 for the variableCensor
indicates that the corresponding value ofTime is a right-censored observation; otherwise, the value is a failure
time. The temperature variableTemp is speci�ed as the independent variable in the model. The MODEL
statement option RELATION=ARR speci�es the Arrhenius relationship.

The option OBSTATS requests statistics computed for each observation in the input data set. The options in
parentheses following OBSTATS indicate which statistics are to be computed. In this case, QUANTILE=.1
.5 .9 speci�es that quantiles of the �tted distribution are to be computed for the value of the variableTemp
at each observation. The CONTROL= option requests quantiles only for those observations in which the
variableCntrl has a value of 1. This eliminates unnecessary quantiles in the OBSTATS table since, in this
case, only the quantiles at the design temperature of130� C are of interest.

The RPLOT, or RELATIONPLOT, statement displays a plot of the lifetime data and the �tted model. The
dependent variableTime, the independent variableTemp, and the censoring indicatorCensor are the same
as in the MODEL statement. The option FIT=MODEL speci�es that the model �tted with the preceding
MODEL statement is to be used for probability plotting and in the relation plot. The option RELATION=ARR
speci�es an Arrhenius scale for the horizontal axis of the relation plot. The PPLOT option speci�es that a
probability plot is to be displayed alongside the relation plot. The type of probability plot is determined by the
distribution named in the DISTRIBUTION statement, in this case, a lognormal (base 10) distribution. Weibull,
extreme value, lognormal (basee), normal, log-logistic, and logistic distributions are also available. The
NOCONF option suppresses the default percentile con�dence bands on the probability plot. The PLOTDATA
option speci�es that the failure times are to be plotted on the relation plot. The PLOTFIT option speci�es
that the 10th, 50th, and 90th percentiles of the �tted relationship are to be plotted on the relation plot. The
options LUPPER and SLOWER specify an upper limit on the life axis scale and a lower limit on the stress
(temperature) axis scale in the plots.

The plots produced by the preceding statements are shown in Figure 18.9. The plot on the left is an overlaid
lognormal probability plot of the data and the �tted model. The plot on the right is a relation plot showing the
data and the �tted relation. The �tted straight lines are percentiles of the �tted distribution at each temperature.
An Arrhenius relation �tted to the data, plotted on an Arrhenius plot, yields straight percentile lines.

Since all the data at150� C are right censored, there are no failures corresponding to150� C on the probability
plot. However, the �tted distribution at150� C is plotted on the probability plot.

Figure 18.9 Probability and Relation Plots for the Class B Insulation Data

The tabular output requested with the MODEL statement is shown in Figure 18.10. The “Model Information”
table provides general information about the data and model. The “Summary of Fit” table shows the number
of observations used, the number of failures and of censored values (accounting for the frequency count), and
the maximum log likelihood for the �tted model.

The “Lognormal Parameter Estimates” table contains the Arrhenius-lognormal model parameter estimates,
their standard errors, and con�dence interval estimates. In this table, INTERCEPT is the maximum likelihood
estimate of� 0, TEMP is the estimate of� 1, and Scale is the estimate of the lognormal scale parameter,� .

Figure 18.10 MODEL Statement Output for the Class B Data

Figure 18.10 continued

The “Observation Statistics” table provides the estimates of the �tted distribution quantiles, their standard
errors, and the con�dence limits. These are given only for the value of130� C, as speci�ed with the
CONTROL= option in the MODEL statement. The predicted median life at130� C corresponds to a quantile
of 0.5, and it is approximately 47,135 hours.

In addition to the MODEL statement output in Figure 18.10, the RELIABILITY procedure produces tabular
output for each temperature that is identical to the output produced with the PROBPLOT statement. This
output is not shown.

Weibull Analysis of Interval Data with Common Inspection Schedule

Table 18.1 shows data for 167 identical turbine parts provided by Nelson (1982, p. 415). The parts were
inspected at certain times to determine which parts had cracked since the last inspection. The times at which
parts develop cracks are to be �tted with a Weibull distribution.

Table 18.1 Turbine Part Cracking Data

Inspection (Months) Number

Start End Cracked Cumulative

0 6.12 5 5
6.12 19.92 16 21
19.92 29.64 12 33
29.64 35.40 18 51
35.40 39.72 18 69
39.72 45.24 2 71
45.24 52.32 6 77
52.32 63.48 17 94
63.48 Survived 73 167

Table 18.1 shows the time in months of each inspection period and the number of cracked parts found in each
period. These data are said to be interval censored since only the time interval in which failures occurred is
known, not the exact failure times. Seventy-three parts had not cracked at the last inspection, which took
place at 63.48 months. These 73 lifetimes are right censored, since the lifetimes are known only to be greater
than 63.48 months.

The interval data in this example are read from a SAS data set with a special structure. All units must have a
common inspection schedule. This type of interval data is calledreadout data. The following SAS program
creates the SAS data set named CRACKS, shown in Figure 18.11, and provides the data in Table 18.1 with
this structure:

data cracks;
input Time units fail;
datalines;

6.12 167 5
19.92 162 16
29.64 146 12
35.4 134 18
39.72 116 18
45.24 98 2
52.32 96 6
63.48 90 17
;

The variableTime is the inspection time—that is, the upper endpoint of each interval. The variableUnits is
the number of unfailed units at the beginning of each interval, and the variableFail is the number of units
with cracks at the inspection time.

Figure 18.11 Listing of the Turbine Part Cracking Data

The following statements use the RELIABILITY procedure to produce the probability plot in Figure 18.12
for the data in the data set CRACKS:

proc reliability data=cracks;
freq fail;
nenter units;
distribution Weibull;
probplot time / readout

ppout
pconfplt
noconf;

run;

The FREQ statement speci�es that the variableFail provides the number of failures in each interval. The
NENTER statement speci�es that the variableUnits provides the number of unfailed units at the beginning of
each interval. The DISTRIBUTION statement speci�es that the Weibull distribution be used for parameter
estimation and probability plotting. The PROBPLOT statement requests a probability plot of the data.

The PROBPLOT statement option READOUT indicates that the data in the CRACKS data set are readout
(or interval) data. The option PCONFPLT speci�es that con�dence intervals for the cumulative probability
of failure be plotted. The con�dence intervals for the cumulative probability are based on the binomial
distribution for time intervals until right censoring occurs. For time intervals after right censoring occurs, the
binomial distribution is not valid, and a normal approximation is used to compute con�dence intervals.

The option NOCONF suppresses the display of con�dence intervals for distribution percentiles in the
probability plot.

Figure 18.12 Weibull Probability Plot for the Part Cracking Data

A listing of the tabular output produced by the preceding SAS statements is shown in Figure 18.13 and
Figure 18.14. By default, the speci�ed Weibull distribution is �tted by maximum likelihood. The line plotted
on the probability plot and the tabular output summarize this �t. For interval data, the estimated cumulative
probabilities and associated con�dence intervals are tabulated. In addition, general �t information, parameter
estimates, percentile estimates, standard errors, and con�dence intervals are tabulated.

Figure 18.13 Partial Listing of the Tabular Output for the Part Cracking Data

Figure 18.13 continued

Figure 18.14 Partial Listing of the Tabular Output for the Part Cracking Data

Figure 18.14 continued

In this example, the number of unfailed units at the beginning of an interval minus the number failing in the
interval is equal to the number of unfailed units entering the next interval. This is not always the case since
some unfailed units might be removed from the test at the end of an interval, for reasons unrelated to failure;
that is, they might be right censored. The special structure of the input SAS data set required for interval data
enables the RELIABILITY procedure to analyze this more general case.

Lognormal Analysis with Arbitrary Censoring

This example illustrates analyzing data that have more general censoring than in the previous example. The
data can be a combination of exact failure times, left censored, right censored, and interval censored data.
The intervals can be overlapping, unlike in the previous example, where the interval endpoints had to be the
same for all units.

Table 18.2 shows data from Nelson (1982, p. 409), analyzed by Meeker and Escobar (1998, p. 135). Each of
435 turbine wheels was inspected once to determine whether a crack had developed in the wheel or not. The
inspection time (in 100s of hours), the number inspected at the time that had cracked, and the number not
cracked are shown in the table. The quantity of interest is the time for a crack to develop.

Table 18.2 Turbine Wheel Cracking Data

Inspection Time Number Number
(100 hours) Cracked Not Cracked

4 0 39
10 4 49
14 2 31
18 7 66
22 5 25
26 9 30
30 9 33
34 6 7
38 22 12
42 21 19
46 21 15

These data consist only of left and right censored lifetimes. If a unit exhibits a crack at an inspection time,
the unit is left censored at the time; if a unit has not developed a crack, it is right censored at the time. For
example, there are 4 left-censored lifetimes and 49 right-censored lifetimes at 1,000 hours.

The following statements create a SAS data set named TURBINE that contains the data in the format
necessary for analysis by the RELIABILITY procedure:

data turbine;
label t1 = �Time of Cracking (Hours x 100 )�;
input t1 t2 f;
datalines;

. 4 0
4 . 39
. 10 4
10 . 49
. 14 2
14 . 31
. 18 7
18 . 66
. 22 5
22 . 25
. 26 9
26 . 30
. 30 9
30 . 33
. 34 6
34 . 7
. 38 22
38 . 12
. 42 21
42 . 19
. 46 21
46 . 15
;

The variablesT1 andT2 represent the inspection times and determine whether the observation is right or
left censored. IfT1 is missing (.), thenT2 represents a left-censoring time; ifT2 is missing,T1 represents a
right-censoring time. The variableF is the number of units that were found to be cracked for left-censored
observations, or not cracked for right-censored observations at an inspection time.

The following statements use the RELIABILITY procedure to produce the probability plot in Figure 18.15
for the data in the data setTURBINE:

proc reliability data = turbine;
distribution lognormal;
freq f;
pplot ( t1 t2 ) / maxitem = 5000

ppout;
run;

The DISTRIBUTION statement speci�es that a lognormal probability plot be created. The FREQ statement
identi�es the frequency variableF. The optionMAXITEM=5000 speci�es that the iterative algorithm that
computes the points on the probability plot takes a maximum of 5,000 iterations. The algorithm does not
converge for these data in the default 1,000 iterations, so the maximum number of iterations needs to be
increased for convergence. The option PPOUT speci�es that a table of the cumulative probabilities plotted
on the probability plot be printed, along with standard errors and con�dence limits.

The tabular output for the maximum likelihood lognormal �t for these data is shown in Figure 18.16.
Figure 18.15 shows the resulting lognormal probability plot with the computed cumulative probability
estimates and the lognormal �t line.

Figure 18.15 Lognormal Probability Plot for the Turbine Wheel Data

Figure 18.16 Partial Listing of the Tabular Output for the Turbine Wheel Data

Figure 18.16 continued

Regression Modeling

This example is an illustration of a Weibull regression model that uses a load accelerated life test of rolling
bearings, with data provided by Nelson (1990, p. 305). Bearings are tested at four different loads, and
lifetimes in106 of revolutions are measured. The data are shown in Table 18.3. An outlier identi�ed by
Nelson (1990) is omitted.

Table 18.3 Bearing Lifetime Data

Load Life (106 Revolutions)

0.87 1.67 2.2 2.51 3.00 3.90 4.70 7.53 14.7 27.76 37.4
0.99 0.80 1.0 1.37 2.25 2.95 3.70 6.07 6.65 7.05 7.37
1.09 0.18 0.2 0.24 0.26 0.32 0.32 0.42 0.44 0.88
1.18 0.073 0.098 0.117 0.135 0.175 0.262 0.270 0.350 0.386 0.456

These data are modeled with a Weibull regression model in which the independent variable is the logarithm
of the load. The model is

� i D � 0 C � 1xi

where� i is the location parameter of the extreme value distribution and

xi D log. load/

for the ith bearing. The following statements create a SAS data set containing the loads, log loads, and
bearing lifetimes:

data bearing;
input load Life @@;
lload = log(load);
datalines;

.87 1.67 .87 2.2 .87 2.51 .87 3.0 .87 3.9

.87 4.7 .87 7.53 .87 14.7 .87 27.76 .87 37.4

.99 .8 .99 1.0 .99 1.37 .99 2.25 .99 2.95

.99 3.7 .99 6.07 .99 6.65 .99 7.05 .99 7.37
1.09 .18 1.09 .2 1.09 .24 1.09 .26 1.09 .32
1.09 .32 1.09 .42 1.09 .44 1.09 .88 1.18 .073
1.18 .098 1.18 .117 1.18 .135 1.18 .175 1.18 .262
1.18 .270 1.18 .350 1.18 .386 1.18 .456
;

Figure 18.17 shows a listing of the bearing data.

Figure 18.17 Listing of the Bearing Data

The following statements �t the regression model by maximum likelihood that uses the Weibull distribution:

ods output modobstats = Residual;
proc reliability data=bearing;

distribution Weibull;
model life = lload / covb

corrb
obstats
;

run;

The PROC RELIABILITY statement invokes the procedure and identi�es BEARING as the input data set.
The DISTRIBUTION statement speci�es the Weibull distribution for model �tting. The MODEL statement
speci�es the regression model, identifyingLife as the variable that provides the response values (the lifetimes)
andLload as the independent variable (the log loads). The MODEL statement option COVB requests the
regression parameter covariance matrix, and the CORRB option requests the correlation matrix. The option
OBSTATS requests a table that contains residuals, predicted values, and other statistics. The ODS OUTPUT
statement creates a SAS data set namedRESIDUAL that contains the table created by the OBSTATS option.

Figure 18.18 shows the tabular output produced by the RELIABILITY procedure. The “Weibull Parameter
Estimates” table contains parameter estimates, their standard errors, and 95% con�dence intervals. In this
table, INTERCEPT corresponds to� 0, LLOAD corresponds to� 1, and SHAPE corresponds to the Weibull
shape parameter. Figure 18.19 shows a listing of the output data setRESIDUAL.

Figure 18.18 Analysis Results for the Bearing Data

Figure 18.18 continued

Figure 18.19 Listing of Data Set Residual

The value of the lifetimeLife and the log loadLload are included in this data set, as well as statistics computed
from the �tted model. The variableXbeta is the value of the linear predictor

x0 O� D O� 0 C Lload O� 1

for each observation. The variableSurv contains the value of the reliability function, the variableSresid
contains the standardized residual, and the variableAresid contains a residual adjusted for right-censored
observations. Since there are no censored values in these data,Sresid is equal toAresid for all the bearings.
See Table 18.32 and Table 18.33 for other statistics that are available in the OBSTATS table and data set. See
the section “Regression Model Statistics Computed for Each Observation for Lifetime Data” on page 1384
for a description of the residuals and other statistics.

If the �tted regression model is adequate, the standardized residuals have a standard extreme value distribution.
You can check the residuals by using the RELIABILITY procedure and theRESIDUAL data set to create an
extreme value probability plot of the residuals.

The following statements create the plot in Figure 18.20:

proc reliability data=residual;
distribution ev;
probplot sresid;

run;

Figure 18.20 Extreme Value Probability Plot for the Standardized Residuals

Although the estimated location is near zero and the estimated scale is near one, the plot reveals systematic
curvature, indicating that the Weibull regression model might be inadequate.

Regression Model with Nonconstant Scale

Nelson (1990, p. 272) and Meeker and Escobar (1998, p. 439) analyzed data from a strain-controlled fatigue
test on 26 specimens of a type of superalloy. The following SAS statements create a SAS data set containing
for each specimen the level of pseudo-stress (Pstress), the number of cycles (in thousands) (Kcycles) until
failure or removal from the test, and a variable to indicate whether a specimen failed (F) or was right censored
(C) (Status):

data alloy;
input pstress kCycles status$ @@;
cen = ( status = �C� );
datalines;

80.3 211.629 F 99.8 43.331 F
80.6 200.027 F 100.1 12.076 F
80.8 57.923 C 100.5 13.181 F
84.3 155.000 F 113.0 18.067 F
85.2 13.949 F 114.8 21.300 F
85.6 112.968 C 116.4 15.616 F
85.8 152.680 F 118.0 13.030 F
86.4 156.725 F 118.4 8.489 F
86.7 138.114 C 118.6 12.434 F
87.2 56.723 F 120.4 9.750 F
87.3 121.075 F 142.5 11.865 F
89.7 122.372 C 144.5 6.705 F
91.3 112.002 F 145.9 5.733 F
;

The following statements �t a Weibull regression model with the number of cycles to failure as the response
variable:

ods output ModObstats = Resids;
proc reliability data = alloy;

distribution Weibull;
model kcycles * cen(1) = pstress pstress * pstress / Relation = Pow Obstats;
logscale pstress;
rplot kcycles * cen(1) = pstress / fit=regression

relation = pow
plotfit 10 50 90
slower=60 supper=160
lupper=500;

label pstress = "Pseudo-Stress";
label kcycles = "Thousands of Cycles";

run;

The data setRESIDS contains standardized residuals created with the ODS OUTPUT statement. The
MODEL statement speci�es a model quadratic in the log of pseudo-stress for the extreme value location
parameter. The quadratic model in pseudo-stress PSTRESS is speci�ed in the MODEL statement, and the
RELATION=POW option speci�es that the log transformation be applied toPstress in the MODEL statement
and the LOGSCALE statement. The LOGSCALE statement speci�es the log of the scale parameter as a linear
function of the log ofPstress. The RPLOT statement speci�es a plot of the data and the �tted regression
model versus the variablePstress. The FIT=REGRESSION option speci�es plotting the regression model

�tted with the preceding MODEL statement. The RELATION=POW option speci�es a log stress axis. The
PLOTFIT option speci�es plotting the 10th, 50th, and 90th percentiles of the regression model at each stress
level. The SLOWER, SUPPER, and LUPPER options control limits on the stress and lifetime axes.

Figure 18.21 displays the parameter estimates from the �tted regression model. Parameter estimates for
both the model for the location parameter and the scale parameter models are shown. Standard errors and
con�dence limits for all parameter estimates are included.

Figure 18.21 Parameter Estimates for Fitted Regression Model

Figure 18.22 displays the plot of the data and �tted regression model.

Figure 18.22 Superalloy Fatigue Data with Fitted Regression Model

The following SAS statements create an extreme values probability plot of standardized residuals from the
regression model shown in Figure 18.23:

proc reliability data = Resids;
distribution ev;
pplot sresid * cen(1) / nofit;

run;

Figure 18.23 Residuals for Superalloy Fatigue Data Regression Model

Regression Model with Two Independent Variables

Meeker and Escobar (1998, p. 447) analyzed data from an accelerated test on the lifetimes of glass capacitors
as a function of operating voltage and temperature. The following SAS statements create a SAS data set
containing the data. There are four lifetimes for each of eight combinations and four censored observations
after the fourth failure for each combination:

data glass;
input Temp Voltage @;
do i = 1 to 4;

cen = 0;
input Hours @; output;

end;
do i = 1 to 4;

cen = 1;
output;

end;
datalines;

170 200 439 904 1092 1105
170 250 572 690 904 1090
170 300 315 315 439 628
170 350 258 258 347 588
180 200 959 1065 1065 1087
180 250 216 315 455 473
180 300 241 315 332 380
180 350 241 241 435 455
;

The following statements analyze the capacitor data. The MODEL statement �ts a regression model with
Temp andVoltage as independent variables. Parameter estimates from the �tted regression model are shown
in Figure 18.24. An interaction term betweenTemp andVoltage is included. The PPLOT statement creates
a Weibull probability plot shown in Figure 18.25 with all temperature-voltage combinations overlaid on
the same plot. The regression model �t is also plotted. The RPLOT statement creates the plot shown in
Figure 18.26 of the data and Weibull distribution percentiles from the regression model as a function of
voltage for values of temperature of 150, 170, and 180:

proc reliability data = glass;
distribution Weibull;
model Hours * cen(1) = temp voltage temp * voltage;
pplot Hours * cen(1) = ( temp voltage ) / fit = model

overlay
noconf
lupper = 2000;

run;

proc reliability data = glass;
distribution Weibull;
model Hours * cen(1) = temp voltage temp * voltage;
rplot Hours * cen(1) = voltage / fit = regression(temp = 150, 170, 180)

plotfit;
run;

Figure 18.24 Parameter Estimates for Fitted Regression Model

Figure 18.25 Probability Plot for Glass Capacitor Regression Model

Figure 18.26 Plot of Data and Fitted Weibull Percentiles for Glass Capacitor Regression Model

Weibull Probability Plot for Two Combined Failure Modes

Doganaksoy, Hahn, and Meeker (2002) analyzed failure data for the dielectric insulation of generator armature
bars. A sample of 58 segments of bars were subjected to a high voltage stress test. Based on examination of
the sample after the test, failures were attributed to one of two modes:

� Mode D (degradation failure): degradation of the organic material. Such failures usually occur later in
life.

� Mode E (early failure): insulation defects due to a processing problem. These failures tend to occur
early in life.

The following SAS statements create a SAS data set that contains the failure data:

data Voltage;
input Hours Mode$ @@;
if Mode = �Cen� then Status = 1;
else Status = 0;
datalines;

2 E 3 E 5 E 8 E 13 Cen 21 E
28 E 31 E 31 Cen 52 Cen 53 Cen 64 E
67 Cen 69 E 76 E 78 Cen 104 E 113 Cen
119 E 135 Cen 144 E 157 Cen 160 E 168 D
179 Cen 191 D 203 D 211 D 221 E 226 D
236 E 241 Cen 257 Cen 261 D 264 D 278 D
282 E 284 D 286 D 298 D 303 E 314 D
317 D 318 D 320 D 327 D 328 D 328 D
348 D 348 Cen 350 D 360 D 369 D 377 D
387 D 392 D 412 D 446 D
;

The variableHours represents the number of hours until a failure, or the number of hours on test if the
sample unit did not fail. The variableMode represents the failure mode: D for degradation failure, E for early
failures, or Cen if the unit did not fail (i.e., is right-censored). The computed variableStatus is a numeric
indicator for censored observations.

The following statements �t a Weibull distribution to the individual failure modes (D and E), and compute
the failure distribution with both modes acting:

proc reliability data=Voltage;
distribution Weibull;
pplot Hours * Status(1) / pref(intersect) = 10

preflabel = (�10th Percentile�)
survtime = 100 200 300 400 500 1000
lupper = 500;

fmode combine = Mode( �D� �E� );
run;

Figure 18.27 contains estimates of the combined failure mode survival function at the times speci�ed with
the SURVTIME= option in the PPLOT statement.

Figure 18.27 Survival Function Estimates for Combined Failure Modes

Figure 18.28 shows Weibull parameter estimates for the two individual failure modes.

Figure 18.28 Parameter Estimates for Individual Failure Modes

Figure 18.29 is a Weibull probability plot of the failure probability distribution with the two failure modes
combined, along with approximate pointwise 95% con�dence limits. A reference line at the 10% point on the
probability axis intersecting the distribution curve shows the tenth percentile of lifetimes when both modes
act to be about 34 hours.

